

# Prüfziffernberechnung für EAN-13 Nummern

Volker Thormählen, 7. August 2021

## Inhalt

1. Aufgabe .....	2
2. Aufbau .....	2
3. Prüfziffernberechnung .....	2
4. Zahlenbeispiel .....	2
5. Rechenvorschrift .....	3
6. VBA-Prozeduren .....	3
7. VBA-Quellcode .....	3
8. Quellen .....	4

## Listings

Listing 1: Quellcode der Subprozedur Pruefziffernberechnung_EAN_13 .....	3
Listing 2: Quellcode der Funktionsprozedur Check_EAN .....	4
Listing 3: Quellcode der Funktionsprozedur NumVal .....	4

## Tabellen

Tabelle 1: Aufbau der EAN-13 Artikelnummer .....	2
Tabelle 2: Zahlenbeispiel zur Prüfziffernberechnung bei EAN 13 [1] .....	2
Tabelle 3: Art, Name und Aufgabe der Prozeduren .....	3

## 1. Aufgabe

In diesen Beitrag wird beschrieben, wie die Gültigkeit der Prüfziffer einer EAN<sup>1</sup>-13 Artikelnummer nach Modulo 10 mit den sich wiederholenden Gewichten 3 und 1 berechnet und überprüft werden kann.

## 2. Aufbau

Die EAN-13 Artikelnummer besteht aus 13 numerischen Datenstellen (s. Tabelle 1). Die Nummerierung der Datenstellen erfolgt von links nach rechts. Die 13. Datenstelle enthält die Prüfziffer. Diese kann aus den anderen 12 Ziffern errechnet werden.

Die ersten 3 Datenstellen beinhalten die EAN-Vorsatznummer. Bei der Konvertierung von ISBN<sup>2</sup> in EAN-13 lautet die Vorsatznummer 978 oder 979.

ISBN ist eine spezielle Nummer für Bücher. Seit der Einführung der ISBN-13 am 1. Januar 2007 ist jede Standardbuchnummer mit der entsprechenden 13-stelligen EAN-Nummer identisch.

An den gewöhnlich durch Bindestriche gegliederten ISBN-13 Buchnummern lassen sich Sprache, Verlags-/Betriebsnummer sowie Titel-/Bandnummer erkennen.

Position	Inhalt
1-3	Präfix, je nach Buch 978 oder 979 (sog. Buchland, engl. bookland)
4	Länder-/Gruppennummer (3 steht für deutschsprachige Bücher)
5-12	Verlags-/Betriebsnummer sowie Titel-/Bandnummer.
13	Prüfziffer

Tabelle 1: Aufbau der EAN-13 Artikelnummer

## 3. Prüfziffernberechnung

Die Prüfziffer einer EAN-13 Nummer wird berechnet, indem die ersten zwölf Ziffern abwechselnd mit 1 und 3 multipliziert und die einzelnen Produkte addiert werden. Die Prüfziffer ist die Differenz der Produktsumme zum nächsten Vielfachen von 10. Falls die Produktsumme durch 10 teilbar ist, ist die Prüfziffer 0.

## 4. Zahlenbeispiel

Eine EAN-13 Nummer *ohne* Gliederungszeichen sei 9783426654538. Die Gewichtung der Datenstellen erfolgt ab Stelle  $z_{12}$ , abwechselnd mit 3 und 1 (s. Tabelle 2). Die Produktsumme beträgt 112. Die Prüfziffer lautet demnach 8.

Datenstelle	$z_1$	$z_2$	$z_3$	$z_4$	$z_5$	$z_6$	$z_7$	$z_8$	$z_9$	$z_{10}$	$z_{11}$	$z_{12}$	$z_{13}$
EAN-13	9	7	8	3	4	2	6	6	5	4	5	3	<b>8</b>
Gewichte	1	3	1	3	1	3	1	3	1	3	1	3	
Produkte	9	21	8	9	4	6	6	18	5	12	5	9	$\Sigma = 112$
Prüfziffernberechnung: $\Sigma \bmod 10 = 2$ $z_{13} = (10 - 2) \bmod 10 = 8$													

Tabelle 2: Zahlenbeispiel zur Prüfziffernberechnung bei EAN 13 [1]

<sup>1</sup> EAN steht für engl. European Article Number

<sup>2</sup> ISBN steht für engl. International Standard Book Number

## 5. Rechenvorschrift

$$z_{13} = \left( 10 - \left( \sum_{i=1}^{12} z_i \cdot 3^{(i+1) \bmod 2} \right) \bmod 10 \right) \bmod 10$$

Das  $(i+1) \bmod 2$  in obiger Rechenvorschrift sorgt für die wechselnde Gewichtung mit 1 und 3.

Quelle: [1].

## 6. VBA-Prozeduren

Die Anwendung besteht aus drei Prozeduren, jeweils programmiert mit der proprietären Skriptsprache *Visual Basic for Applications* (VBA) der Firma *Microsoft* (s. Tabelle 3)

Prozedur		
Art	Name	Aufgabe
Subprozedur	Pruefziffernberechnung_EAN_13	Prüfziffernberechnung für EAN_13
Funktionsprozedur	Check_EAN	Prüfen der übergebenen EAN_13 Artikelnummer auf Gültigkeit.
Funktionsprozedur	NumVal	Alle nicht numerischen Zeichen (wie z. B. Gliederungszeichen) aus der EAN_13 Artikelnummer entfernen.

Tabelle 3: Art, Name und Aufgabe der Prozeduren

## 7. VBA-Quellcode

```
Sub Pruefziffernberechnung_EAN_13()  
    ' Aufgabe: Prüfziffernberechnung für EAN 13 Artikelnummer  
    Dim intLoop As Integer ' Schleifenzähler  
    Dim strEAN As String ' EAN-Nummer  
    Dim varEAN(8) As Variant ' Datenbereich (engl. array)  
    ' Den definierten Datenbereich mit acht EAN-13 Einträgen belegen.  
    ' 5 Nummernteile: Vorsatz, Gruppe, Verlag, Titel-/Bandnummer, Prüfziffer.  
    varEAN(0) = "978-3-426-65453-8" ' Opa, erzähl mal  
    varEAN(1) = "978-3-8273-2394-1" ' Das Access-VBA Codebook  
    varEAN(2) = "978-3-86731-238-7" ' Histamin Irrtum  
    varEAN(3) = "978-1-936561-01-8" ' Nie mehr Parodontose und Karies  
    varEAN(4) = "978-3-446-24645-4" ' Ein ganzes Leben  
    varEAN(5) = "978-3-8362-2638-7" ' jQuery - Das umfassende Handbuch  
    varEAN(6) = "978-3-645-60071-2" ' PowerPoint 2010  
    varEAN(7) = "978-3-455-50236-7" ' Alles über Wikipedia  
    ' Datenbereich mit For-Next-Schleife durchlaufen.  
    For intLoop = 0 To UBound(varEAN) - 1  
        strEAN = varEAN(intLoop)  
        MsgBox "EAN-13-Nummer: " & strEAN & " ist " & _  
            IIf(Check_EAN(strEAN), "richtig", "falsch")  
    Next intLoop  
End Sub
```

Listing 1: Quellcode der Subprozedur *Pruefziffernberechnung\_EAN\_13*

```

Function Check_EAN(strEAN As String) As Boolean
    ' Aufgabe: Prüfen der übergebenen EAN_13 Nummer auf Gültigkeit.
    Dim strNum      As String      ' EAN_13 Artikelnummer ohne Gliederungszeichen
    Dim intLen      As Integer     ' Länge der EAN_13 Artikelnummer
    Dim intLoop     As Integer     ' Schleifenzähler
    Dim intWeight   As Integer     ' Gewichtung mit {3, 1}
    Dim lngSum      As Long        ' Produktsumme der gewichteten Datenstellen
    ' Alle nicht numerischen Zeichen aus strEAN entfernen.
    strNum = NumVal(strEAN)
    intLen = Len(strNum)
    lngSum = 0
    If intLen = 13 Then
        ' Die Gewichtung der Datenstellen erfolgt von rechts nach links.
        ' Rechts wird mit dem Gewicht 3 begonnen, gefolgt von 1 usw.
        intWeight = 3
        For intLoop = intLen - 1 To 0
            lngSum = lngSum + CLng(Mid(strNum, intLoop, 1)) * intWeight
            intWeight = IIf(intWeight = 3, 1, 3)
        Next intLoop
        ' Die Prüfziffer ist richtig, wenn Check_EAN wahr ist.
        Check_EAN = (lngSum Mod 10 = 0)
    ElseIf intLen = 0 Then
        MsgBox "Fehler: Die EAN_13 Nummer ist nicht numerisch!"
        Check_EAN = False
    Else
        MsgBox "Fehler: Die EAN_13 Nummer hat keine 13 Stellen!"
        Check_EAN = False
    End If
End Function

```

Listing 2: Quellcode der Funktionsprozedur *Check\_EAN*

```

Function NumVal(strVal As String) As String
    ' Aufgabe: Alle nicht numerischen Zeichen (sog. Gliederungszeichen) aus strVal
    ' entfernen.
    Dim intLoop As Integer      ' Schleifenzähler
    Dim strChr  As String * 1   ' aktuelles Zeichen
    NumVal = ""                ' Anfangswert
    For intLoop = 1 To Len(strVal)
        strChr = Mid(strVal, intLoop, 1)
        If (Asc(strChr) > 47 And Asc(strChr) < 58) Then
            ' Nur die Ziffern 0 bis 9 übernehmen.
            NumVal = NumVal + strChr
        End If
    Next intLoop
End Function

```

Listing 3: Quellcode der Funktionsprozedur *NumVal*

## 8. Quellen

- [1] o. V., „Formeln zur Berechnung der Prüfziffer,“ 01 08 2021. [Online]. Available: [https://de.wikipedia.org/wiki/Internationale\\_Standardbuchnummer](https://de.wikipedia.org/wiki/Internationale_Standardbuchnummer). [Zugriff am 07 08 2021].